
RESUMO

Modelos unidimensionais, apesar de limitados, são excelentes ferramentas para
estudar princípios básicos de algum sistema físico. Neste documento, as equa-
ções do modelo unidimensional de balanço de energia para o planeta Terra são
descritas e implementadas em Fortran 90. Com o programa em Fortran 90, al-
guns experimentos são realizados com objetivo de mostrar como se comporta,
basicamente, o clima terrestre. Inúmeros outros experimentos podem ser reali-
zados com o modelo. Acredita-se que a disponibilidade do seu código em For-
tran 90 amplie ainda mais a sua utilização por estudantes de graduação e pós-
graduação que desejam iniciar no estudo do clima do nosso planeta.





EXPERIMENTS WITH A ONE-DIMENSIONAL ENERGY BALANCE MODEL

ABSTRACT

One-dimensional models, although limited, are excellent tools to study basic fe-
atures of a physical system. In this document, the equations of one-dimensional
energy balance model for Earth are presented and described, along with a For-
tran 90 program. Some experiments are performed with this model to show basic
Earth’s climate behavior. Several others experiments are possible and it is belie-
ved that the availability of its Fortran 90 code will help a lot of students use it in
their initial Earth’s climate learning.

1 INTRODUÇÃO

Qualquer modelo climático é uma tentativa de simular os muitos processos que
produzem o clima. O objetivo é entender estes processos e prever os efeitos de
mudanças e interações. A simulação é atingida pela descrição do sistema climá-
tico em termos dos princípios físicos, químicos e biológicos básicos. Assim, um
modelo pode ser considerado como um conjunto de equações que descrevem
estas leis.

Os processos do sistema climático ainda não são completamente entendidos,
mas sabe-se de suas complexidades. Isso faz com que os modelos mais comple-
xos já desenvolvidos até hoje produzam somente aproximações dos fenômenos
observados no sistema climático. Além disso, a alta complexidade de modelos
como os Modelos de Circulação Geral (MCG) torna-os inadequados para estu-
dos climáticos do tipo causa-efeito.

A busca por um melhor entendimento desta relações climáticas fez com que ou-
tros tipos de modelos climáticos fossem desenvolvidos. Tais modelos, que tratam
o sistema climático de uma forma mais simplificada, são ferramentas extrema-
mente úteis para o crescimento da compreensão dos processos climáticos e no
desenvolvimento de novas parametrizações e métodos para avaliação da sensi-
bilidade dos modelos mais complexos e realísticos.

Um exemplo desta gama de modelos é o Modelo de Balanço de Energia (MBE).
Este modelo é caracterizado por ser unidimensional (latitude) cuja variável pre-
vista é a temperatura da superfície, obtida em termos de um balanço entre a
radiação solar recebida pelo sistema (INSOLATION, acrônimo de INcoming SO-
LAr radiaTION) e a radiação de onda longa emergente (ROLE).

Os MBE podem ser implementados facilmente em qualquer tipo de linguagem de
programação (FORTRAN, C, BASIC, etc) e na maioria dos computadores consi-
derados de pequeno porte (PC-XT, 386, 486 e superiores). Eles são amplamente
utilizados para investigar a sensibilidade do sistema climático à mudanças exter-
nas e para interpretar os resultados de modelos mais complexos.

A preocupação com a estabilidade do sistema climático à pequenas variações no
aquecimento solar levou Budyko (1969) e Sellers (1969) (independentemente) a
desenvolverem modelos do clima da Terra baseados nas equações do balanço

de calor médio zonal do sistema Terra-Atmosfera.

Uma vez que o balanço de energia é permitido variar em latitude, um termo de
transferência horizontal de energia deve ser introduzido, de modo que a equação
básica para o balanço de energia em cada latitude, φ, é

Cm[∆T (φ)/∆t] = R ↓ (φ)−R ↑ (φ) + F (φ) (1.1)

onde Cm é o calor específico do sistema e pode ser tomado como a inércia
térmica do sistema e R ↓ e R ↑ são os fluxos de entrada e saída de radiação,
respectivamente. A Figura 1.1 mostra como as transferências de energia são
consideradas neste modelo.

Figura 1.1 - Transferências de energia no MBE unidimensional. FONTE: McGuffie e
Henderson-Sellers (1997).

Quando usado para cálculos médios anuais, o albedo em superfície pode ser
mantido constante para cada uma das latitudes. Este tipo de modelo, entretanto,
pode ser usado para cálculos sazonais. Neste caso deve-se permitir variações
do albedo com a temperatura para simular os efeitos de mudanças na extensão
do gelo marinho e de neve.

A dinâmica atmosférica não é modelada neste modelo; é assumido que uma
aproximação de difusão é adequada para a inclusão do transporte de calor. Esta
aproximação relaciona o escoamento de energia diretamente ao gradiente latitu-

dinal de temperatura. Este escoamento é expresso sendo proporcional ao desvio
da temperatura zonal, T, da média global, T .

1.1 As equações do MBE

A equação principal do MBE unidimensional é dada por

Sφ(1− α(Tφ)) = R ↑ (Tφ) + F (Tφ) (1.2)

Fica claro que não é considerado, na Equação 1.2, nenhuma forma de armaze-
namento de energia pelo sistema desde que considera-se o clima em escalas de
tempo onde não há perda ou ganho significativos de energia. Qualquer energia
armazenada implicaria, simplesmente, em um termo adicional, Q(Tφ), no lado
direito da Equação 1.2.

Cada um dos termos na Equação 1.2 é uma função da variável predita Tφ. O al-
bedo da superfície é influenciado pela temperatura e aumentado dramaticamente
quando gelo e neve se formam. A radiação emitida pelo espaço é proporcional
à T 4, embora sobre o intervalo de temperatura de interesse (aprox. 250 até 300
K), esta dependência pode ser considerada linear. O efluxo horizontal da zona
latitudinal é uma função da diferença entre a temperatura daquela zona e a tem-
peratura média global. O termo de armazenamento, uma tentativa para levar em
conta o efeito dos oceanos, depende da diferença entre a temperatura atual e
uma temperatura média de longo prazo. O albedo é descrito por uma função de
passo simples tal que

αφ ≡ α(Tφ)

= 0, 6; quando Tφ ≤ Tc

= 0, 3; quando Tφ > Tc
(1.3)

a qual representa o aumento no albedo na linha de neve; Tc é a temperatura
na linha de neve, está tipicamente entre -10 e 0oC. Devido ao relativo pequeno
intervalo de temperaturas envolvidas, a radiação que deixa a Terra, no topo da
zona de latitude, pode ser aproximada por

Rφ = R ↑ (Tφ) = A+BTφ (1.4)

onde A e B são constantes determinadas empiricamente que levam em conta
o efeito estufa de nuvens, vapor d’água e CO2. A taxa de transporte de energia
pode ser representado como sendo proporcional à diferença entre a temperatura
zonal e a temperatura média global

Fφ = F (Tφ) = K(Tφ − T ) (1.5)

onde K é uma constante empírica.

A substituição das Equações 1.4 e 1.5 na Equação 1.2 forma uma equação que
pode ser rearranjada resultando na equação

Tφ =
Sφ(1− αφ) +KT − A

B +K
(1.6)

Dando-se uma distribuição inicial (first guess) da temperatura e utilizando-se um
esquema adequado de ponderação para distribuir a radiação solar sobre o globo
(por causa da inclinação do eixo da Terra, uma simples distribuição por cosseno
em função da latitude não funciona bem na média anual), aplicações sucessivas
desta equação produzirá, eventualmente, uma solução de equilíbrio. Um curso
alternativo de ação é calcular explicitamente a evolução temporal do modelo cli-
mático incluindo um termo representando a capacidade térmica do sistema. O
primeiro método resulta em resultados computacionalmente mais rápidos en-
quanto que o último permite maiores números de experimentos.

1.2 Implementação do MBE

Com o objetivo de ilustrar a implementação do MBE em uma linguagem de pro-
gramação, é apresentado, nesta seção, o código do MBE em Fortran 90. A imple-
mentação deste modelo foi baseado no apresentado por McGuffie e Henderson-
Sellers (1997), cuja linguagem utilizada foi o BASIC, com pequenas modifica-
ções.

program ener_bal_model

implicit none

integer :: i=0,j=0

real :: Sx=-1.

! DEFINICAO DOS PARAMETROS FIXOS DO MBE E ALGUMAS CONSTANTES

integer, parameter :: MaxIter = 50, NLat = 18

real, parameter :: Tc = -10.0, & ! Temp. formacao de gelo (C)

& A = 204., & ! Const. perda de ROL (W/m2)

& B = 2.17, & ! Const. perda de ROL (W/m2)

& alb_g = 0.62, & ! Albedo c/ gelo

& alb_n = 0.30, & ! Albedo s/ gelo

& K = 3.81, & ! transp. de energia (W/m2.C)

& So = 1370., & ! Const. solar (W/m2)

& pi = 3.1415926 ! PI

! Variaveis e matrizes do modelo

character(len=6), dimension(NLat) :: zonas

real, dimension(NLat) :: Si, & ! Insolation Solar

& Init_T, & ! Temp. inicial

& Lat, & ! Latitudes

& Init_alb, & ! Albedo inicial da superficie

& albedo, & ! Albedo da superficie

& T, & ! Temperatura

& Told, & ! Temperatura da iteracao anterior

& Rin, & ! Radiacao solar entrando em cada lat.

& Rout, & ! Radiacao perdida p/ o espaco

& Tcos, & ! Temperatura x cosseno lat.

& Tfinal ! Temperatura final

real, dimension(MaxIter) :: Tmedia =0 ! Temperatura media global

real :: sumcos=0., &

& TMG=0., & ! Temperatura Media Global

& fracS = 1 ! Fracao da constante solar

! SOLICITACAO DA FRACAO DA CONSTANTE SOLAR AO USUARIO

19



print *,’Insira a fracao da const. solar a ser usada (0-1) : ’

read(*,*) Sx

if(Sx>=0) fracS=Sx

! criando arquivos de saida

open(30,file=’radiacao.out’,status=’unknown’)

open(31,file=’alb_temp.out’,status=’unknown’)

! Prenchendo algumas matrizes

data zonas(1:18)/"80-90N","70-80N","60-70N","50-60N",&

& "40-50N","30-40N","20-30N","10-20N",&

& " 0-10N"," 0-10S","10-20S","20-30S",&

& "30-40S","40-50S","50-60S","60-70S",&

& "70-80S","80-90S"/

data Si(1:18)/0.500,0.531,0.624,0.770,0.892,1.021,&

& 1.120,1.189,1.219,1.219,1.189,1.120,&

& 1.021,0.892,0.770,0.624,0.531,0.500/

data Init_T(1:18)/-16.9,-12.3,-5.1,2.2,8.8,16.2,22.9,&

& 26.1,26.4,26.1,24.6,21.4,16.5,9.9,&

& 2.9,-6.9,-29.5,-42.3/

data Lat(1:18)/85.,75.,65.,55.,45.,35.,25.,15.,5.,&

& -5.,-15.,-25.,-35.,-45.,-55.,-65.,&

& -75.,-85./

data Init_alb(1:18)/0.589,0.544,0.452,0.407,0.357,&

& 0.309,0.272,0.248,0.254,0.241,0.236,&

& 0.251,0.296,0.358,0.426,0.513,0.602,

& 0.617/

! FAZENDO CALCULOS INICIAIS ----> PASSO 1

do i=1,NLat

Rin(i) = So/4*fracS*Si(i)

sumcos = sumcos+cos(Lat(i)*pi/180)

Tcos(i) = Init_T(i)*cos(Lat(i)*pi/180)

Tmedia(1) = Tmedia(1)+Tcos(i)

enddo

Tmedia(1)=Tmedia(1)/sumcos

do i=1,NLat

T(i) = (Rin(i)*(1-Init_alb(i))+K*Tmedia(1)-A)/(B+K)

if( T(i) < Tc ) then

albedo(i) = alb_g

else

albedo(i) = Init_alb(i)

endif

enddo

! FAZENDO CALCULOS INTERMEDIARIOS -> ITERACOES INTERMEDIARIAS

do i=1,MaxIter

do j=1,Nlat

Tcos(j) = T(j)*cos(Lat(j)*pi/180)

Tmedia(i)=Tmedia(i)+Tcos(j)

enddo

Tmedia(i)=Tmedia(i)/sumcos

do j=1,NLat

T(j)=(Rin(j)*(1-albedo(j))+K*Tmedia(i)-A)/(B+K)

if( T(j) < Tc ) albedo(j) = alb_g

enddo

enddo

! PASSO FINAL -> IMPRESSAO DOS RESULTADOS

print *,’Latitude Rin Rout Albedo T’

TMG=Tmedia(MaxIter)

do i=1,NLat

Tfinal(i)=(Rin(i)*(1-Init_alb(i))+K*TMG-A)/(B+K)

if( Tfinal(i) < Tc ) albedo(i) = alb_g

Rout(i)=A+B*Tfinal(i)

! escrevendo na tela

write(*,’(2x,A6,6x,F6.2,4x,F6.2,5x,F5.2,4x,&

& F6.2)’) zonas(i),Rin(i),Rout(i),albedo(i),Tfinal(i)

! escrevendo nos arquivos de saida

write(30,’(F4.0,4x,F6.2,4x,F6.2)’) Lat(i),Rin(i),Rout(i)

write(31,’(F4.0,4x,F5.2,4x,F6.2)’) Lat(i),albedo(i),Tfinal(i)

enddo

print *,’’

print *,’Temperatura media global = ’,TMG

stop

end program ener_bal_model

Os dados de distribuição inicial de temperatura para o globo foram obtidos de
Warren e Schneider (1979), a distribuição da INSOLATION foi obtida de Chylek
e Coakley (1975) e os valores de albedo da superfície foram obtidos de Ellis e
Haar (1976).

Estas distribuições podem ser modificadas de acordo com o tipo de experimento
a ser realizado, sejam elas baseadas em outras fontes observacionais ou por
hipóteses. Deste modo, a utilização do MBE fica sob total controle do usuário
que, com a disponibilização do código, pode modificar todos os parâmetros de
acordo com seus objetivos. É claro, a possibilidade de alteração deste código
não pode ser descartada visto que modificações devido a possíveis erros podem
ser necessárias bem como melhorias.

Uma fácil compreensão do funcionamento deste modelo faz com que outros tipos
de experiências possam ser realizadas com o mesmo. Por exemplo, McGuffie e
Henderson-Sellers (1997) apresentam formas de incluir-se os efeitos das super-
fícies terrestre e oceânicas no MBE.

Espera-se, assim, que a disponibilização do código deste modelo, em Fortran 90,
faça com que os estudantes de modelagem climática, de mudanças climáticas
e de outras áreas afins nos estudos do clima possam compreender melhor os
processos climáticos assim como melhorar este programa.

1.3 Compilação e execução do programa

O programa apresentado na seção anterior pode ser compilado em qualquer
computador que possua um compilador Fortran 90. O código não usa nenhum
comando, subrotina ou função particular a nenhum compilador, usando, portanto,
somente a linguagem padrão do Fortran 90.

A distribuições atuais do Linux dsiponibilizam um compilador Fortran chamado
gfortran. Há a possibilidade deste compilador não acompanhar a instalação pa-
drão de uma determinada distribuição. Mesmo assim é possível obtê-lo por meio
da internet, acessando os repositórios da distribuição usada.

O código pode ser copiado diretamente para um arquivo texto em formato AS-
CII e compilado. Suponha que o arquivo que conterá o código chamar-se-á
mbe1D.f90 e o compilador gfortran será usado. O comando para compilar o có-
digo será

gfortran -o mbe1D.exe mbe1D.f90

que gerará um arquivo executável chamado mbe1D.exe. O nome do arquivo exe-
cutável, resultado da compilação, pode ser escolhido com a opção “-o”, passada
ao compilador. O arquivo executável terá, aproximadamente, 10 Kbytes de tama-
nho, mas pode variar bastante em função do compilador usado.

Finalmente, para executar o programa, basta digitar no terminal

./mbe1D.exe

não esquecendo do “./”, que informa ao sistema operacional que o arquivo local
será executado. Caso não se use estes caracteres o sistema tentará procurar por
algum comando com nome igual ao do arquivo, que, provavelmente, não existe.

O programa imprime os seus resultados na tela e em dois arquivos. No Apêndice
deste trabalho é apresentada a rotina gráfica usada para gerar as Figuras 2.1 e
2.2, que são derivadas destes dois arquivos.

2 Experimentos com o MBE

Neste capítulo algumas experiências feitas com o Modelo de Balanço de Energia
apresentado no capítulo anterior são apresentadas. As experiências são feitas
com o objetivo de verificar a estabilidade do sistema climático simulado pelo MBE
e as consequências no sistema provocadas por algumas espécies de mudanças.

A título de elucidação, os experimentos aqui descritos foram feitos através de
modificações dos parâmetros considerados relevantes. Desta forma, em um
dado experimento, o parâmetro alterado será citado ao passo de que os parâ-
metros restantes são mantidos iguais aos valores pré-definidos por McGuffie e
Henderson-Sellers (1997).

Antes dos experimentos, são apresentadas as distribuições latitudinais da en-
trada de radiação solar (Rin), da perda de radiação de onda longa (ROLE), da
temperatura da superfície e do albedo (Figuras 2.1 e 2.2, respectivamente) para
o caso em que todos as constantes definidas previamente no MBE e os dados
de distribuição de temperatura, albedo e INSOLATION, mencionados no capítulo
anterior, são utilizados.

Figura 2.1 - Variação latitudinal da Radiação Solar (linha contínua) e da ROLE (linha
tracejada) obtida do MBE com os dados médios anuais das distribuições de
INSOLATION, temperatura e albedo.

Pode-se observar claramente a saldo positivo de radiação em quase todo o globo
exceto nas latitudes acima de 800. Estes saldos de radiação refletem nas tem-
peraturas em superfície e nos valores da albedo da superfície exibidos na Figura
2.2. A distribuição de temperatura obtida pelo MBE é simétrica no globo com um
valor relativamente inferior no polo sul. Isto é um resultado da distribuição inicial
de temperatura e albedo (vide programa) que leva em conta a contribuição do
gelo antártico.

Figura 2.2 - Igual a Figura 2.1, exceto para Temperatura e Albedo

Os valores de albedo são praticamente iguais aqueles usados como condições
iniciais, exceto onde a temperatura foi menor que -10oC na qual foi determinada
a ocorrência de gelo (αg = 0, 62).

Apresentado os resultados do MBE com a utilização de todas as condições pre-
estabelecidas por McGuffie e Henderson-Sellers (1997) e as distribuições iniciais
descritas no capítulo anterior, os resultados de alguns experimentos são mostra-
dos a seguir.

O primeiro deles é uma avaliação da sensibilidade do MBE em relação a diminui-
ções na fração constante solar. Para tal, foram utilizados alguns outros valores

para as constantes A, B e K, e analisadas a diminuição da constante solar que
provocaria um congelamento total do globo (Tabela 2.1).

A primeira linha da Tabela 2.1 exibe a diminuição necessária na fração da cons-
tante solar para ocorrência de gelo em todo o globo para os valores pré-definidos
de A, B e K. Desta tabela pode-se observar que o MBE necessita de relativas
pequenas variações na fração da constante solar para os casos em que os co-
eficientes que controlam a perda radiativa são altos. O contrário ocorre quando
estes coeficientes são relativamente baixos, diminuindo a perda radiativa do sis-
tema e mantendo a linha de gelo (latitude na qual o gelo alcança) afastada do
equador para uma diminuição maior da fração da constante solar.

Tabela 2.1 - Variações das constantes empíricas A, B e K, e a resposta do MBE em
função da variação da constante solar para provocar congelamento global.

A B K Resp.
204 2,17 3,81 22%
200 1,50 4,00 19%
200 1,50 2,00 19%
300 3,00 2,00 0%
150 3,00 2,00 54%
150 3,00 5,00 49%

Uma observação mais cuidadosa da Tabela 2.1 evidencia que o MBE não é muito
sensível (em relação a diminuição na fração da constante solar) à variações na
constante que controla os processos de transporte de energia, K. Pode-se ver
que para valores idênticos de A e B, a duplicação de K não altera a quantidade
de diminuição na fração da constante solar para congelar o planeta.

Para exemplificar esta característica do MBE, é apresentado na Figura 2.3 as mu-
danças na temperatura média global em função de K. Fica claro, desta figura, o
já descrito anteriormente. Para uma grande variação positiva de K, a temperatura
média global é praticamente a mesma. Isso só não ocorre quando K aproxima-
se de 0 e de valores menores. Neste caso a temperatura média global tem uma
queda abrupta. Deve-se esclarecer que um K menor que zero violaria uns dos
princípios básicos da termodinâmica pois seria equivalente a uma transferência

de energia da parte mais fria para a parte mais quente, irreal.

Esse comportamento é explicado pelo fato de K pertencer, também, ao denomi-
nador da Equação 1.6. Visto que o valor da constante B é próximo da unidade,
valores negativos de K fazem com que o denominador tenda a zero e que a
Equação 1.6 tenda a −∞.

Apesar do descrito, estas análises da resposta do MBE em função de K não
mostram o efeito físico que valores diferentes de K têm nas distribuições glo-
bais de temperatura e albedo. Na Figura 2.4 são mostrados dois conjuntos de
informações. Um deles, representado pelas linhas com círculos abertos, refere-
se à distribuição de albedo e temperatura quando valores positivos e altos são
atribuídos à variável K. O outro conjunto, representado pelas linhas sem símbo-
los, refere-se à distribuição de albedo e temperatura quando valores positivos e
próximos a zero são atribuídos à K.

Figura 2.3 - Variação da Temperatura Média Global em função de K.

Fica claro, da Figura 2.4, o papel de difusão do excesso de calor recebido nos
trópicos representado pela constante K. Assim, valores altos e positivos de K
provocariam um rápido transporte do calor recebido nos trópicos e ocasionando

uma diminuição na diferença de temperatura entre polo e equador. De maneira
inversa, valores de K próximos a zero ocasionam em um lento transporte de calor
ocasionando em grandes acúmulo de calor nos trópicos e déficit de calor nos po-
los. Em consequência disto, uma grande amplitude da temperatura é observada
na direção equador-polo.

As observações mostram que a superfície terrestre seria totalmente coberta por
neve, durante o inverno, para uma temperatura média anual da superfície de
00C, e os oceanos seriam cobertos de gelo para temperaturas abaixo de -130C.
No presente modelo, a especificação da temperatura na qual a formação de gelo
aconteceria é de -100C. Para avaliar qual seria o comportamento do MBE em fun-
ção da temperatura crítica, TC , este parâmetro foi modicado com os dois valores
citados acima. A Figura 2.5 mostra as distribuições de temperatura e albedo para
TC = 00C (linhas com círculo aberto) e para TC = −130C (linhas sem símbolos).

Figura 2.4 - Igual a Figura 2.2, exceto quando a K → ∞ (linhas com círculo aberto) e
quando K → 0 (linhas sem símbolos). As linhas tracejadas referem-se a
temperatura (eixo y esquerdo) e as linhas contínuas referem-se ao albedo
(eixo y direito).

Na Figura 2.5 vê-se que a modificação na TC não proporciona grande variação
nas distribuições de temperatura e albedo. Entretanto, a sensibilidade do MBE

em relação a diminuição na fração da constante solar aumenta no caso em que
TC = 00C. Neste caso, é necessário somente uma diminuição de 12% na fração
da constante solar para que todo o globo seja coberto por gelo. Já quando TC =

−130C esta diminuição deve ser da ordem de 25% para provocar este efeito.

Além disto, sabe-se que o valor do albedo do gelo depende de outros fatores tais
como a própria condição do gelo, da cobertura de nuvens e do tipo de vegetação
sobre a qual o gelo se encontra. Para testar a resposta do MBE à diferentes valo-
res de αg, valores entre 0,5 e 0,8 foram atribuídos a este parâmetro. A Tabela 2.2
mostra a porcentagem de queda na fração da constante solar necessária para
a ocorrência de congelamento global, em função de αg. A temperatura média
global, T , também é exibida.

Figura 2.5 - Igual a Figura 2.2, exceto quando a TC é igual a 00C (linhas com círcu-
los abertos) e igual a −100C (linhas sem símbolos). As linhas tracejadas
referem-se a temperatura (eixo y esquerdo) e as linhas contínuas referem-
se ao albedo (eixo y direito).

Esta faixa de variação de αg, sugerida por McGuffie e Henderson-Sellers, não
provoca grandes modificações, em relação ao valor pré-definido (αg = −100C).
Apenas uma pequena queda na temperatura média global é observada com o
aumento de αg. Isto deve-se a menores quantidade de radiação solar absorvida

quando o gelo, com altos valores de albedo, está presente.

Tabela 2.2 - Variação no valor do albedo do gelo (αg) e as respectivas temperaturas mé-
dias globais (T ) e porcentagem de diminuição da constante solar para ocor-
rência de congelamento global.

αg T Resp.
0,5 289,4 25%
0,6 288,9 22%
0,7 288,4 19%
0,8 287,9 17%

As constantes A e B definidas na Equação 1.4 determinam a perda radiativa do
planeta para o espaço. Existem diversos valores definidos por diferentes autores.
Budyko (1969) usou A=202 Wm−2 e B=1,45 Wm−2 0C−1, enquanto Cess (1976)
utilizou A=212 Wm−2 e B=1,6 Wm−2 0C−1. A despeito de valores em particu-
lar, valores diferentes provocam mudanças na sensibilidade do MBE bem como
no clima do mesmo. Menores valores para estas constantes ocasionariam taxas
menores de perda de energia do planeta e, consequentemente, maiores valores
para a temperatura média global, T . Ao contrário, maiores valores proporciona-
riam maiores perdas radiativas e menores temperaturas médias globais.

O descrito acima pode ser observado na Tabela 2.1 onde diversos valores para
as constantes A e B são mostrados. Pode-se notar, por exemplo, que com A=200
Wm−2 e B=1,5 Wm−2 0C−1 uma diminuição de 19% na fração da constante solar
é requerida para o congelamento global total enquanto que com A=300 Wm−2

e B=3,0 Wm−2 0C−1 nenhuma diminuição na fração solar é necessária para que
haja o congelamento total.

Para exemplificar este fato, fez-se uma análise da variação de B em função da
temperatura média global, T (Figura 2.6). É claro o comportamento assintótico
deste parâmetro, ou seja, para B � 1, T se aproxima de 0oC, e para valores
próximos a zero, T →∞.

Segundo McGuffie e Henderson-Sellers (1997), estas constantes incorporam ao
sistema climático do MBE o efeito estufo devido às nuvens, ao vapor d’água e
ao CO2. Visto que relações proporcionais são observadas entre a quantidade de

vapor d’água e CO2, e a temperatura média global, resta especular a respeito do
efeito das nuvens.

Figura 2.6 - Igual a fig. 2.3, exceto para variação de B.

O albedo planetário é duplicado de 15 a 30%, e a emissão de onda longa é
reduzida em aproximadamente 30 Wm−2 devido às nuvens. Por causa do cance-
lamento parcial entre estes dois efeitos, o efeito resultante das nuvens no fluxo
global de energia radiativa para o planeta é uma redução de aproximadamente
20 Wm−2 (HARTAMANN, 1994). Além disso, é sugerido que para nuvens baixas e
médias o efeito de albedo é mais importante que o efeito estufa da nuvem, en-
quanto que as nuvens cirrus, por exemplo, têm um impacto muito menor sobre
o albedo, de modo que, seu efeito geral é aquecer o sistema pelo aumento do
efeito estufa.

Infelizmente, nosso entendimento de como a quantidade global e o tipo das nu-
vens são determinados é atualmente insuficiente, e as estimativas da magnitude
e do sinal do processo de realimentação das nuvens são, basicamente, desco-
nhecidos. Isso faz com que este processo seja uma das maiores incertezas em
estudos de sensibilidade climática.

Em virtude do exposto acima e da resposta do modelo frente a uma variação
em B, uma possível explicação para o comportamento da temperatura média
global em função de B seria que este parâmetro representaria o processo de
realimentação de nuvens no sentido negativo, ou seja, em relação ao albedo das
nuvens. Apresentado desta forma, B seria diretamente proporcional à cobertura
de nuvens baixas e médias, que alteram o albedo terrestre.

3 CONCLUSÕES

Nos capítulos precedentes foram apresentados o Modelo de Balanço de Energia,
as suas equações, o código em Fortran 90 e alguns experimentos realizados com
o mesmo.

Apesar da simplicidade do modelo e pelo fato de não levar em conta muitos
dos processos pertencentes ao sistema climático, ele tem grande utilidade na
avaliação da dependência do clima nos mecanismos de transporte de calor para
o pólo, das eras glaciais e dos problemas envolvendo o acoplamento oceano-
atmosfera.

Nos experimentos aqui apresentados, os diversos parâmetros do MBE foram
analisados e suas respectivas influências no clima e em sua sensibilidade. A
sensibilidade do modelo foi avaliada de duas maneiras: por meio da variação
da temperatura média global, T , e da diminuição na fração da constante solar
requerida para o congelamento global ocorrer.

Notou-se que o MBE aumenta a sua sensibilidade à variações na fração da cons-
tante solar quando mudanças são impostas às constantes que controlam a perda
de energia do planeta para o espaço. Grandes variações positivas e negativas
nas constantes A e B ocasionam, respectivamente, um aumento e uma diminui-
ção na sensibilidade do MBE.

Modificações no valor correspondente ao albedo do gelo, αg, e na temperatura
crítica, TC , na qual ocorre a formação de gelo, não provocam grandes variações
na sensibilidade do MBE. Tanto analisados em função da temperatura média
global, T , quanto sob a perspectiva da diminuição na fração da constante solar,
estes parâmetros não indicam grandes variações em resposta as modificações
nos parâmetros relativos ao gelo.

Dois parâmetros, em particular, foram analisados de forma mais detalhada. Os
parâmetros B, o qual é um dos coeficientes que controlam a perda radiativa do
planeta, e K, o qual controla o transporte de calor para os pólos, apresentaram
comportamentos assintóticos quando analisados em função de T . Isso não é
nenhuma surpresa devido ao fato de que ambos coeficientes encontram-se no
denominador da equação preditora da temperatura da superfície.

Conhecido o comportamento matemático destes coeficientes, há a necessidade
de ligá-las ao mundo real, a uma situação física que explique este comporta-
mento. No caso do coeficiente K, a relação é direta e simples. Quanto maior for
este coeficiente maior é a eficácia do transporte meridional de calor. O contrário
ocorre para valores que tendem a zero. Isto é ilustrado na Figura 2.4.

Entretanto, quanto a variável B, não há uma análise simples e direta. Este co-
eficiente, juntamente com A, controla a perda radiativa do planeta, ou seja, o
efeito estufa devido ao CO2, vapor d’água e nuvens. Como as relações entre
quantidade de CO2 e vapor d’água, e a temperatura média global são direta-
mente proporcionais, resta sugerir que o comportamento de B indica mudanças
na cobertura de nuvens médias e baixas, que influenciam no albedo planetário.
Deve-se deixar claro que a falta de conhecimentos básicos sobre os processos
de retroalimentação permite que incertezas a respeito desta afirmação persis-
tam.

A facilidade com que diversos tipos de experimentos podem ser realizados faz
deste modelo uma das ferramentas mais conhecidas em estudos de modelagem
climática. Como mencionado anteriormente, outros componentes do sistema cli-
mático podem ser incluídos neste modelo. McGuffie e Henderson-Sellers (1997)
sugerem outros tipos de parametrizações e experimentos mais complexos (com
mais componentes climáticos) com o MBE.

Cursos que têm como um de seus objetivos um conhecimento básico sobre os
processos climáticos têm muito a ganhar com a disponibilidade do código do
MBE em uma linguagem de programação bastante difundida no meio acadêmico,
Fortran. A possibilidade de modificação e a grande quantidade de experimentos
que podem ser criados para este modelo são pontos fortes para a sua utilização
no ensino da física básica do nosso planeta.

A APÊNDICE - ROTINAS GRÁFICAS

Neste capítulo adicional é apresentada a rotina gráfica para a criação das Figu-
ras 2.1 e 2.2. Estas figuras foram geradas por meio do NCAR Command Lan-
guage (NCL), uma linguagem interpretada criada, especificamente, para análise
e plotagem de dados científicos, pelo National Center for Atmospheric Research
(NCAR). Está disponível gratuitamente no site http://www.ncl.ucar.edu.

Os dois arquivos gerados pelo programa mbe1D.f90 são carregados por esta
rotina e seus dados plotados, gerando dois arquivos gráficos no formato PNG.

O leitor deste documento é convidado a visitar o site do NCL para melhor com-
preender os comandos e a estrutura do script apresentado abaixo.

load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_code.ncl"

load "$NCARG_ROOT/lib/ncarg/nclscripts/csm/gsn_csm.ncl"

begin

; nome dos arquivos de entrada e saida

entrada1 = "radiacao.out"

entrada2 = "alb_temp.out"

saida = "resultados_mbe"

tipoGraf = "ps"

; le arquivos de saida do Modelo de Balanco de Energia

radiacao = asciiread( entrada1, (/18,3/), "float" )

tempAlbedo = asciiread( entrada2, (/18,3/), "float" )

; ambiente grafico

wks = gsn_open_wks( tipoGraf, saida )

; recursos do primeiro grafico

resXY = True

resXY@gsnMaximize = True

resXY@xyLineThicknesses = 3.

resXY@tiYAxisString = "radiac~H-12F35~K~F21H2~a"+\

"~H-14F35~D~F21H3~o [W m~S~-2~N~]"

resXY@tiXAxisString = "latitude"

resXY@tmXTOn = False

resXY@tmYROn = False

; plota grafico da variacao latitudinal de Rin e Rout

plotRad = gsn_csm_xy( wks, radiacao(:,0), \

transpose(radiacao(:,1:2)), resXY )

; recursos para o segundo grafico

; ... para o eixo Y esquerdo

resL = True

resL@gsnMaximize = True

resL@xyLineThicknesses = 3.

resL@tiYAxisString = "albedo"

resL@tiXAxisString = "latitude"

; ... para o eixo Y direito

resR = True

resR@xyDashPatterns = 1

resR@xyLineThicknesses = 3.

resR@tiYAxisString = "temperatura [~S~o~N~C]"

; plota grafico da variacao latitudinal de T e albedo

plotTA = gsn_csm_xy2( wks, tempAlbedo(:,0), tempAlbedo(:,1), \

tempAlbedo(:,2), resL, resR )

; fecha ambiente grafico

delete( wks )

; converte figuras para png

system( "convert -geometry 1000x1000 -density 300 -trim "+ \

saida+"."+tipoGraf+" grafs.png" )

; apaga arquivo PS

system( "rm -f "+saida+"."+tipoGraf )

end

Uma vez instalado o NCL (o seu site possui todas as informações necessárias
para este procedimento), basta executar a seguinte linha de comando para rodar
o script acima:

ncl plotaGraficos_MBE.ncl

assumindo que o script seja salvo num arquivo chamado plotaGraficos_MBE.ncl
e que o modelo tenha sido rodado com sucesso, gerando os arquivos de saída
necessários para a geração dos gráficos.

Aos leitores fica o convite para tentarem reproduzir experimentos realizados com
o MBE e apresenados neste documento, bem como os outros gráficos mostra-
dos.

